-----------------------------------------------

This code comes from Bradley D. Brown's Oracle 8i Web Development, Chapter  25

-----------------------------------------------

-----------------------------------------------

<meta http-equiv=’pragma’ content=’no-cache’> 

-----------------------------------------------

-----------------------------------------------

set serveroutput on
execute your_web_procedure
execute owa_util.show_page

-----------------------------------------------

-----------------------------------------------

rem locks.sql - shows all locks in the database.
set linesize 132
set pagesize 60

column object         heading 'Database|Object' format a15 truncate
column lock_type      heading 'Lock|Type'       format a4 truncate
column mode_held      heading 'Mode|Held'       format a15 truncate
column mode_requested heading 'Mode|Requested'  format a15 truncate
column sid            heading 'Session|ID'
column username       heading 'Username'        format a20 truncate
column image          heading 'Active Image'    format a20 truncate

spool locks.lis

select c.sid,
       substr(object_name,1,20) OBJECT,
       c.username,
       substr(c.program,length(c.program)-20,length(c.program)) image,
       decode(b.type,
             'MR', 'Media Recovery',
             'RT', 'Redo Thread',
             'UN', 'User Name',
             'TX', 'Transaction',
             'TM', 'DML',
             'UL', 'PL/SQL User Lock',
             'DX', 'Distributed Xaction',
             'CF', 'Control File',
             'IS', 'Instance State',
             'FS', 'File Set',
             'IR', 'Instance Recovery',
             'ST', 'Disk Space Transaction',
             'TS', 'Temp Segment',
             'IV', 'Library Cache Invalidation',
             'LS', 'Log Start or Switch',
             'RW', 'Row Wait',
             'SQ', 'Sequence Number',
             'TE', 'Extend Table',
             'TT', 'Temp Table',
              b.type) lock_type,
       decode(b.lmode,
              0, 'None',           /* Mon Lock equivalent */
              1, 'Null',           /* NOT */
              2, 'Row-SELECT (SS)',     /* LIKE */
              3, 'Row-X (SX)',     /* R */
              4, 'Share',          /* SELECT */
              5, 'SELECT/Row-X (SSX)',  /* C */
              6, 'Exclusive',      /* X */
              to_char(b.lmode)) mode_held,
       decode(b.request,
              0, 'None',           /* Mon Lock equivalent */
              1, 'Null',           /* NOT */
              2, 'Row-SELECT (SS)',     /* LIKE */
              3, 'Row-X (SX)',     /* R */
              4, 'Share',          /* SELECT */
              5, 'SELECT/Row-X (SSX)',  /* C */
              6, 'Exclusive',      /* X */
              to_char(b.request)) mode_requested
from   sys.dba_objects a, sys.v_$lock b, sys.v_$session c
where  a.object_id = b.id1 and b.sid = c.sid 
and    owner not in ('SYS','SYSTEM');

-----------------------------------------------

-----------------------------------------------

oerr ora 100

-----------------------------------------------

-----------------------------------------------

create or replace function web_total (in_order in integer) 
return integer 
is
  nbt_running_total integer := 0;
  cursor order_detail_cur (in_id in integer) is
    select qty, amount
    from   order_detail
    where  order_id = in_id;
begin
  for order_detail_rec in order_detail_cur(in_order) loop
    nbt_running_total := nbt_running_total + 
                         floor(order_detail_rec.qty * 
                         order_detail_rec.amount);
  end loop;
  update order_header
  set    order_total = nbt_running_total
  where  order_id    = in_order;
  return nbt_running_total;
end web_total;

-----------------------------------------------

-----------------------------------------------

create or replace function web_total (in_order in integer) 
return integer 
is
  nbt_running_total integer := 0;
  cursor order_detail_cur (in_id in integer) is
    select qty, amount
    from   order_detail
    where  order_id = in_id;
begin
  for order_detail_rec in order_detail_cur(in_order) loop
    nbt_running_total := nbt_running_total + 
                         floor(order_detail_rec.qty * 
                         order_detail_rec.amount);
  end loop;
  update order_header
  set    order_total = nbt_running_total
  where  order_id    = in_order;
  return nbt_running_total;
exception
  when value_error then
    raise_application_error(-20000, 'Max order total exceeded.');  
end web_total;

-----------------------------------------------

-----------------------------------------------

create or replace function web_total (in_order in integer) 
return integer 
is
  nbt_ora_error     varchar2(4000);
  nbt_running_total integer        := 0;
  cursor order_detail_cur (in_id in integer) is
    select qty, amount
    from   order_detail
    where  order_id = in_id;
begin
  for order_detail_rec in order_detail_cur(in_order) loop
    nbt_running_total := nbt_running_total + 
       floor(order_detail_rec.qty * order_detail_rec.amount);
  end loop;
  update order_header
  set    order_total = nbt_running_total
  where  order_id    = in_order;
  return nbt_running_total;
exception
  when value_error then
    raise_application_error(-20000, 'Max order total exceeded.');
  when others then
    nbt_ora_error := sqlerrm;
    raise_application_error(-20000, nbt_ora_error);  
end web_total;

-----------------------------------------------

-----------------------------------------------

Create or replace procedure user_error_message
  (in_type    number,
   in_message varchar2) is
begin
  htp.htmlOpen;
  if in_type = ‘U’ then
    htp.title(‘Support Required – Unexpected Error’);
    htp.header(1,’Unexpected Error, Please Call Support’);
    htp.header(3,‘Please contact I.T. support at extension 2043’);
  else
    htp.title(in_message);
    htp.header(1,in_message);
    htp.print(‘Please press the BACK button and fix the problem’);
  end if;
end user_error_message;

-----------------------------------------------

-----------------------------------------------

create or replace function web_total (in_order in integer) 
return integer 
is
  nbt_ora_error     varchar2(4000);
  nbt_running_total integer        := 0;
  cursor order_detail_cur (in_id in integer) is
    select qty, amount
    from   order_detail
    where  order_id = in_id;
begin
  for order_detail_rec in order_detail_cur(in_order) loop
    nbt_running_total := nbt_running_total + 
      floor(order_detail_rec.qty * order_detail_rec.amount);
  end loop;
  update order_header
  set    order_total = nbt_running_total
  where  order_id    = in_order;
  return nbt_running_total;
exception
  when value_error then
    user_error_message(‘E’, 'Max order total exceeded.');  
  when others then
    nbt_ora_error := sqlerrm;
    user_error_message(‘U’, nbt_ora_error);  
end web_total;

-----------------------------------------------

-----------------------------------------------

-- DDL creates the system_error table.
create table system_error(
    system_error_id        number(10),
    program_unit_name      varchar2(4000),
    unit_type              varchar2(4000),
    unit_owner             varchar2(4000),
    execution_location     varchar2(4000),
    browser                varchar2(4000),
    server_software        varchar2(4000),
    server_name            varchar2(4000),
    server_port            varchar2(4000),
    remote_host            varchar2(4000),
    remote_addr            varchar2(4000),
    path                   varchar2(4000),
    script_name            varchar2(4000),
    oracle_error_text      varchar2(4000),
    additional_information varchar2(4000),
    call_stack             varchar2(4000),
    error_stack            varchar2(4000),
    insert_time            date,
    insert_user            varchar2(30))
tablespace any_big_one
storage   (initial 1M next 100K pctincrease 0);

-----------------------------------------------

-----------------------------------------------

-- This DDL creates the sequence number for the primary key 
-- values for the system_error table
create sequence system_error_id
       increment by 1
       start     with 1
       maxvalue  100000
       minvalue  1
       cycle
       cache     100
       order;

-----------------------------------------------

-----------------------------------------------

-- Name: log_error
-- Description: Records a error in the error logging table.
-- Syntax: log_error (in_location,
--                    in_error,
--                    in_text,
--                    in_commit)
--
-- Where: in_location   = The reference to a physical location within
--                        the procedure/function in which the error 
--                        occurred.
--        in_error      = The Oracle error message.
--        in_text       = Any additional information provided by the
--                        developer to aid in identify the problem.
--                        For example, this might be a rowid/account
--                        number.
--        in_commit     = Boolean flag to determine if this procedure
--                        should perform a commit after writing to the
--                        error table.
create or replace procedure log_error (
                     in_location in varchar2 default NULL,
                     in_error    in varchar2 default NULL,
                     in_text     in varchar2 default NULL,
                     in_commit   in boolean  default TRUE) 
is
  pu_failure          exception;
  pragma              exception_init (pu_failure, -20000);
  nbt_ora_error       varchar2(4000);
  nbt_id              number;
  nbt_program_unit    varchar2(4000);
  nbt_unit_type       varchar2(4000);
  nbt_unit_owner      varchar2(4000);
  nbt_unit_line       number;
  nbt_call_stack      varchar2(4000) := 
                      substr(dbms_utility.format_call_stack, 1, 4000);
  nbt_error_stack     varchar2(4000) := 
                      substr(dbms_utility.format_error_stack, 1, 4000);
  nbt_browser         varchar2(4000) := 
                      owa_util.get_cgi_env('HTTP_USER_AGENT');
  nbt_server_software varchar2(4000) := 
                      owa_util.get_cgi_env('SERVER_SOFTWARE');
  nbt_server_name     varchar2(4000) := 
                      owa_util.get_cgi_env('SERVER_NAME');
  nbt_server_port     varchar2(4000) := 
                      owa_util.get_cgi_env('REMOTE_HOST');
  nbt_remote_host     varchar2(4000) := 
                      owa_util.get_cgi_env('REMOTE_ADDR');
  nbt_remote_addr     varchar2(4000) := 
                      owa_util.get_cgi_env('SERVER_PORT');
  nbt_script_name     varchar2(4000) := 
                      owa_util.get_cgi_env('SCRIPT_NAME');
  nbt_path            varchar2(4000) := 
                      owa_util.get_owa_service_path;

  cursor sys_err_id_cur is
    select system_error_id.nextval
    from   dual;

begin
  open  sys_err_id_cur;
  fetch sys_err_id_cur into nbt_id;
  close sys_err_id_cur;

  owa_util.who_called_me(nbt_unit_owner, nbt_program_unit, 
                         nbt_unit_line, nbt_unit_type); 

  -- If the pseudo column user isn’t the “user” that you want, you can
  -- retrieve the correct user now. For example, if your application 
  -- is using a cookie to store the real user or if you’re using basic
  -- or digest authentication, you could retrieve the username 
  -- at this point.

  insert into system_error 
  (
system_error_id, program_unit_name, unit_type, unit_owner,

execution_location, browser, server_software, server_name,
    server_port, remote_host, remote_addr, path,
    script_name, oracle_error_text, additional_information, 
    call_stack, error_stack, insert_time, insert_user) 
  values
(
nbt_id, nbt_program_unit, nbt_unit_type, nbt_unit_owner,

in_location, nbt_browser, nbt_server_software, nbt_server_name,
    nbt_server_port, nbt_remote_host, nbt_remote_addr, nbt_path,
    nbt_script_name, in_error, in_text, 
    nbt_call_stack, nbt_error_stack, sysdate, user);

  if in_commit then
    commit;
  end if;

  -- Display a message to the user now – we could send an email to 
  -- someone, page an administrator, send a message through DBMS_PIPE
  nbt_ora_error := sqlerrm;
  user_error_message(‘U’, nbt_ora_error);  

exception
  when others then
    raise pu_failure;
end log_error;

-----------------------------------------------

-----------------------------------------------

create or replace function web_total (in_order in integer) 
return integer 
is
  nbt_code_section  varchar2(4000);
  nbt_ora_error     varchar2(4000);
  nbt_running_total integer;
  cursor order_detail_cur (in_id in integer) is
    select qty, amount
    from   order_detail
    where  order_id = in_id;
begin
  nbt_code_section := 'Beginning';
  nbt_running_total   := 0;
  for order_detail_rec in order_detail_cur(in_order) loop
    nbt_code_section  := 'In cursor loop';
    nbt_running_total := nbt_running_total + 
                         floor(order_detail_rec.qty * 
                         order_detail_rec.amount);
  end loop;
  nbt_code_section    := 'Before update';
  update order_header
  set    order_total = nbt_running_total
  where  order_id    = in_order;
  nbt_code_section    := 'After update';
  return nbt_running_total;
exception
  when value_error then
    user_error_message('E', 'Max order total exceeded.');  
  when others then
    rollback;
    log_error (nbt_code_section,
               sqlerrm,
               'Runtime error calculating total for Order '||
                  to_char(in_order),
               true);
end web_total;

-----------------------------------------------

-----------------------------------------------

create or replace procedure web_payroll_calc(in_emp     in number,
                                  out_gross  out number,
                                  out_ded    out number,
                                  out_net    out number,
                                  out_error  out varchar2) is
  nbt_timesheet_found boolean       := false;
  nbt_code_section    varchar2(4000);

  cursor timesheet_cur (in_id in number) is
    select *
      from  web_timesheet
      where emp_id         = in_id
      and   processed_flag = 'N';
begin
  --Access all outstanding timesheet entries for the employee
  --and process. 
  nbt_code_section := 'Before timesheet_rec loop'; 
  for timesheet_rec in timesheet_cur(in_emp) loop
    nbt_code_section := 'In timesheet_rec loop'; 
    nbt_timesheet_found := true;
    --Call incremental payroll logic here…
    …
  end loop;
  nbt_code_section := 'After timesheet_rec loop'; 

  --Prevent check creation if no timesheet records were found.
  if not nbt_timesheet_found then
    out_error := 'Check not created - no timesheet records found.';
  end if;
exception
  when others then
    rollback;
    log_error (nbt_code_section,
               sqlerrm,
               'Runtime error doing payroll calculation for emp #'||to_char(in_emp),
               true);
end web_payroll_calc;

-----------------------------------------------

-----------------------------------------------

create or replace procedure accept_emp_info
  (in_emp varchar2) is
nbt_gross_amt    number(10,2);
nbt_ded_amt      number(10,2);
nbt_net_amt      number(10,2);
nbt_error_msg    varchar2(4000);
nbt_code_section varchar2(4000);
begin
  
  -- Perform payroll calculations and populate the payroll tables.
  -- The gross, deduction, and net check amounts will be returned
  -- use these to complete the payroll check record and display 
  -- to the browser
  web_payroll_calc(to_number(in_emp), nbt_gross_amt,
                   nbt_ded_amt, nbt_net_amt, nbt_error_msg);

  --Check for errors in the called package.
  if nbt_error_msg is not null then
     user_error_message('E', nbt_error_msg);  
     return;  -- Stops the procedure here
  end if;

  htp.htmlOpen;
  -- This is where we would display all the information we calculated 
  -- back to the browser. This would simply use htp.prints, 
  -- tableOpen,…
  …
  htp.htmlClose;

  --Commit transaction
  commit;

exception
  when others then
    rollback;
    log_error (nbt_code_section,
               sqlerrm,
               'Runtime error doing payroll calculation for emp #'||in_emp,
               true);
end accept_emp_info;

-----------------------------------------------

-----------------------------------------------

create or replace procedure web_payroll_calc(in_emp     in number,
                                  out_gross  out number,
                                  out_ded    out number,
                                  out_net    out number,
                                  out_error  out varchar2) is
  no_timesheet      exception;
  nbt_timesheet_found boolean       := false;
  nbt_code_section    varchar2(4000);

  cursor timesheet_cur (in_id in number) is
    select *
      from  web_timesheet
      where emp_id         = in_id
      and   processed_flag = 'N';
begin
  --Access all outstanding timesheet entries for the employee
  --and process. 
  nbt_code_section := 'Before timesheet_rec loop'; 
  for timesheet_rec in timesheet_cur(in_emp) loop
    nbt_code_section := 'In timesheet_rec loop'; 
    nbt_timesheet_found := true;
    --Call incremental payroll logic here…
    …
  end loop;
  nbt_code_section := 'After timesheet_rec loop'; 

  --Prevent check creation if no timesheet records were found.
  if not nbt_timesheet_found then
    raise no_timesheet; 
  end if;
exception
  when no_timesheet then
    raise_application_error(-20000, 'Check not created - no timesheet 
                                     records found.');
  when others then
    rollback;
    log_error (nbt_code_section,
               sqlerrm,
               'Runtime error doing payroll calculation for emp #'||to_char(in_emp),
               true);
end web_payroll_calc;

-----------------------------------------------

-----------------------------------------------

create or replace procedure accept_emp_info
(in_emp number) is
-- User-defined exception to prevent exception raised in 
-- stored program units (by raise_application_error) from being 
-- handled by others clause.
pu_failure       exception;
pragma           exception_init(pu_failure, -20000);
nbt_gross_amt    number(10,2);
nbt_ded_amt      number(10,2);
nbt_net_amt      number(10,2);
nbt_error_msg    varchar2(4000);
nbt_code_section varchar2(4000);
begin
  
  -- Perform payroll calculations and populate the payroll tables.
  -- The gross, deduction, and net check amounts will be returned
  -- use these to complete the payroll check record and display
  --  to the browser
  web_payroll_calc(in_emp, nbt_gross_amt, nbt_ded_amt, nbt_net_amt, nbt_error_msg);

  -- This is where we would display all the information we calculated
  -- back to the browser. This would simply use htp.prints, tableOpen
  htp.htmlOpen;
  …
  htp.htmlClose;

  --Commit transaction
  commit;

exception
  when PU_FAILURE then
    user_error_message('E', sqlerrm);  
  when others then
    rollback;
    log_error (nbt_code_section,
               sqlerrm,
               'Runtime error doing payroll calculation for emp #'||to_char(in_emp),
               true);
end accept_emp_info;

-----------------------------------------------

-----------------------------------------------

create or replace package web_debug is
  debug_mode boolean;
  start_time integer;
  procedure init;
  procedure write_log(in_location varchar2,in_debug_text in varchar2);
  procedure really_write_log(in_location varchar2,
                             in_debug_text in varchar2);
end web_debug;
/

create or replace package body web_debug is
  procedure init is
    -- We could store the debug value in a table, but for this
    -- example, to shorten the code, we’re just going to hardcode
    -- this. It’s important to put this in the package body
    -- so that the value can be changed, the package body 
    -- recompiled without invalidating all referencing procedures.
  begin
    web_debug.debug_mode := true;
    if web_debug.start_time is NULL then
       web_debug.start_time := dbms_utility.get_time;
    end if;
  end init;

  procedure write_log
    (in_location varchar2,
     in_debug_text in varchar2) is
  begin
    -- Only write the log, if we are in debug mode
    web_debug.init;
    if web_debug.debug_mode then
       really_write_log(in_location, in_debug_text);
    end if;
  end write_log;

  procedure really_write_log
    (in_location varchar2,
     in_debug_text in varchar2) is
  nbt_log_time        date           := sysdate;
  nbt_start_time      integer        := dbms_utility.get_time; 
  nbt_program_unit    varchar2(4000);
  nbt_unit_type       varchar2(4000);
  nbt_unit_owner      varchar2(4000);
  nbt_unit_line       number;
  nbt_browser         varchar2(4000) := 
                      owa_util.get_cgi_env('HTTP_USER_AGENT');
  nbt_server_software varchar2(4000) := 
                      owa_util.get_cgi_env('SERVER_SOFTWARE');
  nbt_server_name     varchar2(4000) := 
                      owa_util.get_cgi_env('SERVER_NAME');
  nbt_server_port     varchar2(4000) := 
                      owa_util.get_cgi_env('REMOTE_HOST');
  nbt_remote_host     varchar2(4000) := 
                      owa_util.get_cgi_env('REMOTE_ADDR');
  nbt_remote_addr     varchar2(4000) := 
                      owa_util.get_cgi_env('SERVER_PORT');
  nbt_script_name     varchar2(4000) := 
                      owa_util.get_cgi_env('SCRIPT_NAME');
  nbt_path            varchar2(4000) := owa_util.get_owa_service_path;
  begin
    owa_util.who_called_me(nbt_unit_owner, nbt_program_unit, 
                         nbt_unit_line, nbt_unit_type); 

    -- Insert the date and time of the log entry, the time the
    -- grand (the one called from the user’s browser) procedure 
    -- started (in 100s of a second), the time this call was made
    -- (in 100s of a second) and the debug text.
    insert into debug_log
      (log_time, grand_start_time, debug_start_time, 
       debug_text, program_unit_name, unit_type, unit_owner,
       execution_location, browser, server_software, server_name,
       server_port, remote_host, remote_addr, path,
       script_name, log_user) 
    values
      (nbt_log_time, web_debug.start_time, nbt_start_time, 
       in_debug_text, nbt_program_unit, nbt_unit_type, nbt_unit_owner,
       in_location, nbt_browser, nbt_server_software, nbt_server_name,
       nbt_server_port, nbt_remote_host, nbt_remote_addr, nbt_path,
       nbt_script_name, user);
  exception
    when others then
      raise_application_error(-20000, 'Error in write_log procedure');
  end really_write_log;
end web_debug;

-----------------------------------------------

-----------------------------------------------

create or replace procedure web_submit
  (in_emp_no    integer,
   in_name      varchar2,
   in_address   varchar2,
   in_address_2 varchar2,
   … ) is
  insert_counter integer;
  update_counter integer;
  nbt_code_section varchar2(4000);
begin
  nbt_code_section := 'Procedure web_submit starting';
  web_debug.write_log(nbt_code_section, 'Incoming Args – Emp No: '||
                      in_emp_no||…);

  nbt_code_section := 'Validation';
  web_debug.write_log(nbt_code_section,’Starting’);
  web_validate(in_emp_no, …);

  --Begin calculation routines.
  nbt_code_section := 'Calculation';
  web_debug.write_log(nbt_code_section,’Starting’);
  web_calc(in_emp_no, …);

  --Create transaction records.
  nbt_code_section := 'Transaction Creation';
  web_debug.write_log(nbt_code_section,’Starting’);

  insert into order_header...
  web_debug.write_log(nbt_code_section,'Order Header: '||
     sql%rowcount||' records inserted.');

  insert into order_detail...
  web_debug.write_log(nbt_code_section,'Order Detail: '||
     sql%rowcount||' records inserted.');

  update inventory...
  web_debug.write_log(nbt_code_section,' Inventory: '||
     sql%rowcount||' records inserted.');

  nbt_code_section := 'Procedure web_submit ending normally';
  web_debug.write_log(nbt_code_section,NULL);
exception
  when others then
    rollback;
    log_error (nbt_code_section,
               sqlerrm,
               'Processing emp #'||to_char(in_emp_no),
               true);
end web_submit;

-----------------------------------------------

-----------------------------------------------

create or replace package web_validation is
  type error_type is table of varchar2(4000) index by binary_integer;
  io_error_array web_validation.error_type;
  procedure check_customer (
            in_cust        in     number,
            in_total       in     number,
            io_error_array in out web_validation.error_type);
  procedure check_stock (
            in_item        in owa_util.ident_arr, 
            in_qty_wanted  in owa_util.ident_arr,
            io_error_array in out web_validation.error_type);
end web_validation;
/

create or replace package body web_validation is
  -- This procedure validates the customer account.
  procedure check_customer (
    in_cust        in     number,
    in_total       in     number,
    io_error_array in out web_validation.error_type) is
    nbt_cust_found boolean := false;
    cursor cust_cur(in_id in number) is
      select credit_amt
      from   web_cust
      where  cust_id = in_id;
  begin
    for cust_rec in cust_cur (in_cust) loop
      nbt_cust_found := true;
      if cust_rec.credit_amt < in_total then
        io_error_array(io_error_array.count + 1) :=
            'Your credit limit ('||cust_rec.credit_amt||
            ') is insufficient to complete order - '||
            'please reduce items and/or quantities.';
      end if;
    end loop;
    if not nbt_cust_found then
      io_error_array(io_error_array.count + 1) :=
           'Incorrect account number - please re-enter.';
    end if;
  exception
    when others then
      io_error_array(io_error_array.count + 1) :=
       'Unable to access your account. Please call customer service.';
  end check_customer;

  --This procedure verifies that the item is still available.
  procedure check_stock (
    in_item        in owa_util.ident_arr, 
    in_qty_wanted  in owa_util.ident_arr,
    io_error_array in out web_validation.error_type) is
    nbt_item_found boolean;
    cursor stock_cur(in_id in number) is
      select on_hand
      from   web_stock
      where  item_id = in_id;
  begin
    for nbt_item_no in 1 .. in_item.count loop
      nbt_item_found := false;
      for stock_rec in stock_cur (in_item(nbt_item_no)) loop
        nbt_item_found := true;
        if stock_rec.on_hand < in_qty_wanted(nbt_item_no) then
          io_error_array(io_error_array.count + 1) :=
               'Only '||stock_rec.on_hand||' units of item '||
               in_item(nbt_item_no)||
               'are available - please re-enter quantity.';
        end if;
      end loop;
      if not nbt_item_found then
        io_error_array(io_error_array.count + 1) :=
             'Incorrect item number - please re-enter.';
      end if;
    end loop;
  exception
    when others then
      io_error_array(io_error_array.count + 1) :=
       'Unable to verify inventory. Please call customer service.';
  end check_stock;
end web_validation;

-----------------------------------------------

-----------------------------------------------

create or replace procedure web_process_order
  (in_cust  number,
   in_total number,
   in_item  owa_util.ident_arr,
   in_qty   owa_util.ident_arr) is
  done             exception;
  io_error_array   web_validation.error_type;
  nbt_code_section varchar2(4000);
begin
  --Validate the customer.
  nbt_code_section := 'Validate customer';
  web_validation.check_customer(in_cust, in_total, io_error_array);

  --Validate the selected items.
  nbt_code_section := 'Checking Stock';
  web_validation.check_stock(in_item, in_qty, io_error_array);

  --If any of the validation procedures encountered errors then
  --put all error messages in a Web page and send back to the user.
  --Otherwise, process the order.
  if io_error_array.count > 0 then
    for loopctr in 1 .. io_error_array.count loop
      user_error_message('E', io_error_array(loopctr));
    end loop;
    raise done;
  end if;

  --Code to actually process order should be placed here.
  --The same error array should be used to record processing errors
  --and present them to the user.
  …
exception
  when done then
    null;
  when others then
    rollback;
    log_error (nbt_code_section,
               sqlerrm,
               'Runtime error processing order for cust #' ||
                 to_char(in_cust)…,
               true);
end web_process_order;

-----------------------------------------------

-----------------------------------------------

create or replace package my_package is
  -- Always maintain version when making coding changes
  version varchar2(4000) := '19991230.1';
  …
end my_package;

-----------------------------------------------

-----------------------------------------------

select text
from   user_source
where  name = ‘MY_PACKAGE’
and    type = ‘PACKAGE’
and    upper(text) like ‘%VERSION%’;

-----------------------------------------------

-----------------------------------------------

-- Always maintain version when making coding changes
version varchar2(4000) := '19991230.1';

-----------------------------------------------

-----------------------------------------------

create or replace package my_package is
  version varchar2(4000);
  function what_version return varchar2;
  pragma restrict_references (what_version, WNDS);
  …
end my_package;

-----------------------------------------------

-----------------------------------------------

create or replace package body my_package is
  function what_version return varchar2 is
  begin
    -- Always maintain version when making coding changes
    my_package.version := '19991230.1';
    return version;
  end what_version;
end my_package;

-----------------------------------------------

-----------------------------------------------

create or replace procedure check_versions
  (in_program_unit      varchar2 default '%') is
   nbt_dml_statement   varchar2(4000);
   nbt_package_version varchar2(4000);
   nbt_record_count    integer;
   nbt_version_cursor  integer;
  cursor source_cur is
    select distinct name
    from   user_source
    where  type    = 'PACKAGE'
    and    name like in_program_unit;
cursor object_cur (in_object varchar2) is
    select to_char(last_ddl_time, 'YYYYMMDD') last_ddl_time
    from   user_objects
    where  object_name = in_object
    and    object_type = 'PACKAGE';
begin
  htp.htmlOpen;
  htp.headOpen;
  htp.title('View Program Unit Versions');
  htp.headClose;
  htp.bodyOpen;
  htp.tableOpen('BORDER=1');
  htp.tableRowOpen;
  htp.tableHeader('Program Unit');
  htp.tableHeader('Version');
  htp.tableHeader('Last DDL Date');
  htp.tableHeader('Notes');
  htp.tableRowClose;
  for source_rec in source_cur loop
    nbt_version_cursor := dbms_sql.open_cursor;
    nbt_dml_statement := 'select '||source_rec.name||
                         '.what_version from dual';
    begin
      dbms_sql.parse(nbt_version_cursor, nbt_dml_statement, 
                     dbms_sql.native);
      dbms_sql.define_column(nbt_version_cursor, 1, 
                     nbt_package_version, 100);
      nbt_record_count := dbms_sql.execute(nbt_version_cursor);
      if dbms_sql.fetch_rows(nbt_version_cursor) > 0 then
          dbms_sql.column_value(nbt_version_cursor, 1, 
                     nbt_package_version);
      else
        nbt_package_version := 'Version Reporting Failed';
      end if;
    exception
      when others then
        nbt_package_version := 'Version Reporting Not Supported';
    end;
    htp.tableRowOpen;
    htp.tableData(lower(source_rec.name));
    htp.tableData(nbt_package_version);
    for object_rec in object_cur (source_rec.name) loop
      htp.tableData(object_rec.last_ddl_time);
      if substr(nbt_package_version,1,8) != 
                object_rec.last_ddl_time then
         htp.tableData('*** Versions do not match ***');
      else
         htp.tableData;
      end if;
    end loop;
    htp.tableRowClose;
    dbms_sql.close_cursor(nbt_version_cursor);
  end loop;
  htp.tableClose;
  htp.bodyClose;
  htp.htmlClose;
end check_versions;

-----------------------------------------------

